Modeling Firms

Short Course on CGE Modeling, United Nations ESCAP

John Gilbert

Professor
Department of Economics and Finance
Jon M. Huntsman School of Business
Utah State University
jgilbert@usu.edu

July 22-26, 2013
Now that we understand the formulation of demand in a typical CGE model, we turn to the question of supply.

We’ll start with the basic cost minimization problem, and then turn to the problem of production in a competitive economy.

In the process, we’ll begin working with the CES function, which is widely used in constructing various components of CGE models.
Session Outline

1. The cost minimization problem
2. Example using CES
3. Building the model in GAMS
   - Setting up the model
   - Calibration
   - Simulation and testing
Suppose that the firm uses inputs of labor ($L$) and capital ($K$), for which it must pay market prices $w$ and $r$.

Its technology is described by the production function $q = q(K, L)$. This function represents the relationship between inputs and the maximum output that can be produced, and is assumed to be continuous and to exhibit diminishing returns to each factor and CRTS.

The firm seeks to minimize its expenditure for a given level of output, $\bar{q}$. 
The problem can be written:

$$\min \; \mathcal{L} = rK + wL + \lambda [\bar{q} - q(K, L)]$$

The first order conditions are:

$$\frac{\partial \mathcal{L}}{\partial K} = r - \lambda \frac{\partial q}{\partial K} = 0$$
$$\frac{\partial \mathcal{L}}{\partial L} = w - \lambda \frac{\partial q}{\partial L} = 0$$
$$\frac{\partial \mathcal{L}}{\partial \lambda} = \bar{q} - q(K, L) = 0$$
The solution to the minimization problem is the simultaneous solution to these three equations for $K$, $L$ and $\lambda$, expressed in terms of $w$, $r$ and $\bar{q}$.

At an optimum, each factor price is equal to the value of the marginal product of that factor.

Solving explicitly for the optimal input bundles yields the firm’s factor demand functions.
Geometric Interpretation

\[ K \]

\[ \frac{C^*}{r} \]

\[ K^* \]

\[ L^* \]

\[ \bar{q} \]

\[ \frac{C^*}{w} \]
Suppose the firm’s technology can be described by the production function \( q = \gamma[\delta K^\rho + (1 - \delta)L^\rho]^{1/\rho} \), where \( \rho \leq 1 \) and \( \rho \neq 0 \). The factor demands will be:

\[
K = \frac{\bar{q}\delta^\sigma r^{-\sigma}}{\gamma \left[(\delta r^{-\rho})^\sigma + ((1 - \delta)w^{-\rho})^\sigma\right]^{1/\rho}}
\]

\[
L = \frac{\bar{q}(1 - \delta)^\sigma w^{-\sigma}}{\gamma \left[(\delta r^{-\rho})^\sigma + ((1 - \delta)w^{-\rho})^\sigma\right]^{1/\rho}}
\]

where \( \sigma = 1/(1 - \rho) \) — the elasticity of substitution.
Set Notation

Let the set of factors be $J = \{K, L\}$. We can then let the factor demands be represented by $F_j$, and the factor prices by $r_j$. Using this notation, we can write the factor demands compactly as:

$$F_j = \frac{\bar{q}\delta^\sigma_j r_j^{-\sigma}}{\gamma \left[ \sum_{\forall k \in J} (\delta_k r_k^{-\rho})^\sigma \right]^{\frac{1}{\rho}}} \quad \forall j \in J$$

Note how the summation term is defined over $k$ rather than $j$. Expenditure can be rewritten:

$$E = \sum_{\forall j \in J} r_j F_j$$

This is the form of the model that we will use in our GAMS program.
Now let us consider exactly how the problem can be expressed in the GAMS language.

Our first task is to create a set which will index the factors:

```
SET J Factors /K,L/;
ALIAS (J, JJ);
```

The keyword ALIAS defines another set called JJ that has exactly the same elements as J.
Next, we need to define labels for all of the parameters and exogenous variables in the model. We are also going to define labels for the initial values of our endogenous variables:

PARAMETERS
GAMMA Shift parameter in production
DELTA(J) Share parameters in production
RHO Elasticity parameter in production
Q Output level
R(J) Factor prices
EO Initial expenditure
FO(J) Initial factor use levels;
Our next task is to assign names for the variables:

VARIABLES
E Expenditure level
F(J) Factor use levels;
We enter names for equations in the model in the same way that we enter names for parameters and variables:

EQUATIONS
EXPENDITURE Expenditure function
FDEMAND(J) Factor demand functions;

Then we define the structure of the equations in terms of the variables and parameters:

EXPENDITURE..E=E=SUM(J, R(J)*F(J));
FDEMAND(J)..F(J)=E=(Q/GAMMA)*DELTA(J)**(1/(1-RHO))*
                   R(J)**(-1/(1-RHO))/SUM(JJ, (DELTA(JJ)*
                   R(JJ)**(-RHO))**(1/(1-RHO)))**(1/RHO);

These are the GAMS equivalents of the equations we derived.
With the labels created, we can calibrate:

\[
\begin{align*}
R(J) &= 1; \\
Q &= 100; \\
FO(J) &= 50; \\
EO &= \text{SUM}(J, R(J) \times FO(J)); \\
RHO &= 0.1; \\
DELTA(J) &= \frac{R(J) / FO(J)^{(RHO-1)}}{\text{SUM}(JJ, R(JJ) / FO(JJ)^{(RHO-1)}}; \\
GAMMA &= \frac{Q}{(\text{SUM}(J, DELTA(J) \times FO(J)^{RHO}))^{(1/RHO)}};
\end{align*}
\]
Next we set the starting values:

\[ E.L = E_0; \]
\[ F.L(J) = F_0(J); \]

and assign lower bounds:

\[ F.LO(J) = 0; \]
The last stage is to tell GAMS which of our equations constitute the model, in this case all of them, and then run a test solution:

MODEL PRODUCTION /ALL/;
SOLVE PRODUCTION USING NLP MINIMIZING E;
Exercises

- We have now built our second simple numerical simulation model. Once we are satisfied that the model is correctly replicating the solution, we can examine the effects of changes in the economic situation by altering the values of parameters and/or exogenous variables and executing another SOLVE command.

- Some things to try:
  - Changing factor prices and/or expenditure in various combinations.
  - Altering the model parameters.
  - Experimenting with different functional forms.
Further Reading

- Most of this material is covered in intermediate microeconomics texts such as Varian (2009).
- Dixit (1990) is a good introduction to optimization.
- This GAMS example is developed more fully in Gilbert and Tower (2013), chapter 4.